# Maintainer: Alexander F Rødseth <xyproto@archlinux.org>
# Maintainer: Morten Linderud <foxboron@archlinux.org>
# Contributor: Thomas Dziedzic <gostrc@gmail.com>
# Contributor: Chris Baker <baker.chris.3@gmail.com>

pkgname=python-nltk
pkgver=3.8.1
pkgrel=2
pkgdesc='Natural language processing in Python'
arch=('any')
url='https://www.nltk.org/'
license=('APACHE')
depends=('python-yaml' 'python-click' 'python-regex')
makedepends=('python-setuptools')
checkdepends=('python-nose' 'nltk-data' 'python-requests' 'python-regex')
optdepends=('nltk-data: test data'
	    'python-numpy: used for calculations'
	    'python-matplotlib: used for plotting')
source=($pkgname-$pkgver.tar.gz::"https://github.com/nltk/nltk/archive/$pkgver.tar.gz")
sha256sums=('676970e2b7aa0a7184e68f76e0c4f2756fd1b82559a509d5656a23117faeb658')

build() {
  cd nltk-$pkgver
  python setup.py build
}

# Need to fix the nltk-data split
#check(){
#  cd nltk-$pkgver
#  python ./nltk/test/runtests.py
#}

package() {
  cd nltk-$pkgver
  python setup.py install --root="$pkgdir" --optimize=1 --skip-build
}
